SET ISTRUZIONI DEL PIC


· TIPO : CARICAMENTO E SPOSTAMENTO DATI
· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	MOVLW
	K
	MOVE LITERAL TO W
	INSERISCE NEL REGISTRO W  (ACCUMULATORE ) IL VALORE INSERITO AL POSTO DI K



· ESEMPIO
	     Org     00h
Start

     Movlw     174

     -  -  -  -
	Dopo aver eseguito questo programma l’accumulatore conterrà un byte uguale a 174 cioè in binario 10101110.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	MOVWF

	f
	MOVE W TO F
	TRASFERISCE DA W ( ACCUMULATORE ) AL F (FILE REGISTER)  O ALL’INDIRIZZO RAM DEL REGISTRO


· ESEMPIO
	     MOVWF     32
	IN QUESTO CASO TRASFERISCE W NELL’INDIRIZZO 32 DELLA RAM

	     Mowlw     10H 

     ;scrive nel reg. W il valore 10H
     movwf     01H

     ;e lo memorizza all’ind. 01H cioè al TMR0
	CARICA 10H ALL’INDIRIZZO 01H CHE CORRISPONDE AL REGISTRO TMR0 NEL BANCO 0.

	     Movlw     10H

     ;scrive nel reg. W il valore 10H

     movwf     TMR0

     ;e lo memorizza nel reg. TMR0
	CARICA 10H ALL’INDIRIZZO TMR0, LA DEFINIZIONE E L’INDIRIZZO SONO PRELEVATE DAL FILE P16F877..INC CARICATO ALL’INIZIO DEL PROG. DA INCLUDE    “P16F877.INC”.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	MOVF

	f,d
	MOVE F
	COPIA IL CONTENUTO DEL REG. NEL W O NEL REG. STESSO
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	     Movf    PIPPO,W
     Movf    PIPPO,F
	CARICA IN W (ACCUMULATORE) IL VALORE DEL REGISTRO PIPPO;
CARICA IN PIPPO IL SUO STESSO VALORE, APPARENTEMENTE NON SERVE MA ANDANDO AD AGIRE SUL FLAG Z, POSSIAMO VEDERE SE IL CONTENUTO DEL REGISTRO E’ ZERO O DIVERSO DA 0.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	SWAPF


	f,d
	SWAP NIBBLES IN F
	SCAMBIA I 4 BIT PIU’ SIGNIFICATIVI CON I 4 MENO SIGNIFICATIVI DEL REGISTRO F

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC 


· ESEMPIO
	     Mowlw     147
     Movwf     pippo

     Swapf     pippo,f
	CARICA 147 NELL’ACCUMULATORE W=10101110;
LO METTE NEL REGISTRO PIPPO PIPPO=10101110;

SWAPPA I NIBBLES DI PIPPO PIPPO=11101010.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z = 1
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	CLRF


	f
	CLEAR F
	PONE A ZERO TUTTI I BIT DEL REGISTRO SPECIFICATO


· ESEMPIO
	     CLRF     01h

       CLRF     TMR0
	IN QUESTO MODO AZZERIAMO IL REGISTRO CON INDIRIZZO 01H CHE CORRISPONDE AL TMR0;
AZZERIAMO IL TMR0 SE E’ STATO USATO INCLUDE “P16F877.INC”


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z=1
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	CLRW


	
	CLEAR W
	PONE A ZERO TUTTI I BIT DEL REGISTRO W (ACCUMULATORE)


· ESEMPIO
	
	


· TIPO : ARITMETICHE
· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z , C , DC
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	ADDLW


	K
	ADD LITERAL TO W
	SOMMA IL CONTENUTO DI W ALLA COSTANTE K E PONE IL RISULTATO IN W 


· ESEMPIO
	     Org 00H

Start

       Movlw     10

       Addlw      12

       -  -  -  -
	DOPO AVER ESEGUITO QUESTO PROGRAMMA L’ACCUMULATORE VALE 22


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z , C , DC
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	ADDWF


	f,d
	ADD W AND F
	SOMMA IL VALORE DI W CON QUELLO DEL REG. SPECIFICATO E PONE IL RISULTATO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC 


· ESEMPIO
	     Add1    equ    0CH
       Add2     equ    0DH

       Org     00H

;primo addendo = 10

       movlw     10

       movwf     add1

;secondo addendo = 15

      movlw     15

      movwf     add2

;W=add1

     movf     add1,W

;W = W + add2

     addwf    add2,W
	QUESTO PROGRAMMA FA LA SOMMA TRA DUE REGISTRI ED INFLUENZA I FLAG Z, DC E C;
Z VALE 1 SE IL RISULTATO E’ 0;

DC VALE 1 SE IL RISULTATO DELL’OPERAZIONE E’ UN NUMERO SUPERIORE A 15;

C VALE 1 SE IL RISULTATO E’ POSITIVO, CIOE’ IL BIT 7 DEL RISULTATO VALE 0, VALE 0 SE IL RISULTATO E’ NEGATIVO E IL BIT 7 VALE 1. 




· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z , C , DC
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	 SUBLW


	K
	SUBTRACT LITERAL FROM W
	SOTTRAE  W DALLA COSTANTE K E LA SCRIVE IN W


· ESEMPIO
	        Org     00H
Start

        Movlw     15

        Sublw      10       
	QUESTO PROGRAMMA SOTTRAE DALL’ACCUMULATORE  ( W=15) IL VALORE DELLA COSTANTE  E METTE IL RISULTATO NELL’ACCUMULATORE


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z , C , DC
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	SUBWF

	f,d
	SUBTRACT W FROM F
	SOTTRAE DAL REGISTRO IL CONTENUTO DI W E IL RISULTATO VIENE POSTO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	INCF

	f,d
	INCREMENT F
	INCREMENTA IL REGISTRO SPECIFICATO E PONE IL RISULTATO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	DECF


	f,d
	DECREMENT F
	DECREMENTA IL CONTENUTO DELL’INDIRIZZO O DEL REGISTRO SPECIFICATO E  PONE IL RISULTATO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	     Movlw     23H
     Movwf     0CH

     Decf     OCH,F
	;scrive in W il valore 23H
;copia nel registro 0CH il valore di W

;decrementa il valore contenuto nel registro 0CH


· TIPO : LOGICHE
· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	ANDLW


	K
	AND LITERAL WITH W
	EFFETTUA L’AND BIT A BIT TRA L’ACCUMULATORE E LA COSTANTE K, IL RISULTATO VIENE POSTO IN W


· ESEMPIO
	     Org     00H
Start

     Movlw     10101010b

     Andlw     11110000b

     -  -  -  -
	DOPO AVER ESEGUITO QUESTO PROGRAMMA L’ACCUMULATORE W VARRA’ 10100000b


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	ANDWF

	f,d
	AND W WITH F
	EFFETTUA L’AND BIT A BIT TRA W E IL REGISTRO SPECIFICATO E PONE IL RISULTATO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	       Movlw     01010101b
       Movwf     0CH

       Movlw     00001111b

       Andwf     0CH,W 
	CARICA NELL’ACCUMULATORE 01010101b;
MEMORIZZA ALLA LOCAZIONE 0CH IL VALORE DI W (01010101b);
CARICA IN W LA STRINGA 00001111b;
EFFETTUA L’AND TRA IL CONTENUTO DELLA LOCAZIONE 0CH E W E IL RISULTATO VIENE POSTO IN W;
QUESTO PROGRAMMA CREA UNA MASCHERA PER ESTRARRE SOLO I QUATTRO BIT MENO SIGNIFICATIVI DI UNA STRINA POSTA IN UN REGISTRO O UNA LOCAZIONE RAM.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	IORLW


	K
	INCLUSIVE OR LITERAL WITH W
	ESEGUE L’OR TRA LA COSTANTE E W, IL RISULTATO E’ POSTO IN W


· ESEMPIO
	     Org     00H

Start

     Movlw     00001111b

     Iorlw    11110000b

     -  -  -  -
	DOPO AVER ESEGUITO QUESTO PROGRAMMA L’ACCUMULATORE W VARRA’ 11111111b


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	IORWF

	f,d
	INCLUSIVE OR W WITH F
	ESEGUE L’OR (BIT A BIT) TRA W E IL REGISTRO O LOCAZIONE SPECIFICATA E PONE IL RISULTATO IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	XORLW


	K
	EXCLUSIVE OR LITERAL WITH W
	EFFETTUA L’OR ESCLUSIVO BIT A BIT TRA LA COSTANTE E W, PONE IL RISULTATO IN W


· ESEMPIO
	
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	XORWF

	f,d
	EXCLUSIVE OR W WITH F
	EFFETTUA L’OR ESCLUSIVO TRA REG. O LOCAZIONE E L’ACCUMULATORE (W) , PONE IL RISULTATO IN :

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : Z
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	COMF


	f,d
	COMPLEMENT F
	FA IL COMPLEMENTO DEL REGISTRO E LO PONE IN :

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


· ESEMPIO
	
	


· TIPO : ROTAZIONE E SET/RESET DI SINGOLI BIT
· DATO : BYTE
· FLAG DI STATO MODIFICATI : C
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	RLF


	f,d
	ROTATE LEFT THROUGH CARRY
	IL CONTENUTO DEL REGISTRO F VIENE TRASLATO A SX, IL BIT 7 VA NEL FLAG CARRY E IL CONTENUTO DEL FLAG C VA NEL BIT 0, IL RISULTATO VA IN :
d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC


[image: image1.png]]
2

= ster T





· ESEMPIO
	
	PUO’ ESSERE UTILIZZATO IN TRASMISSIONI E RICEZIONI SERIALI, O PER MOLTIPLICARE O DIVIDERE PER DUE.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : C
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	RRF


	f,d
	ROTATE RIGHT THROUGH CARRY
	IL CONTENUTO DEL REGISTRO F VIENE TRASLATO A DX, IL BIT 0 VA NEL FLAG CARRY E IL CONTENUTO DEL FLAG C VA NEL BIT 7, IL RISULTATO VA IN : 

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC

· ESEMPIO
	
	PUO’ ESSERE UTILIZZATO IN TRASMISSIONI E RICEZIONI SERIALI, O PER MOLTIPLICARE O DIVIDERE PER DUE.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	BCF


	f,b
	BIT CLEAR F
	PONE A “0” IL BIT b DEL REGISTRO f ( I BIT VANNO DAL BIT 0 AL BIT 7


· ESEMPIO
	     BSF     portB,0

     BCF     portB,0
	GENERA UN IMPULSO POSITIVO SUL PIN RB0 DELLA DURATA DI 1 μs SE SI HA UN CLOCK DI 4 MHz.
	


· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	BSF


	f,b
	BIT SET F
	PONE A “1” IL BIT b DEL REGISTRO f ( I BIT VANNO DAL BIT 0 AL BIT 7


· ESEMPIO
	     BSF     portB,0
     BCF     portB,0
	GENERA UN IMPULSO POSITIVO SUL PIN RB0 DELLA DURATA DI 1 μs SE SI HA UN CLOCK DI 4 MHz.


· TIPO : CONTROLLO FLUSSO, SALTI E SOBROUTINE (PARTE 1)
· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	BTFSC

	f,b
	BIT TEST F, SKIP IF CLEAR 
	TESTA IL BIT b DEL REGISTRO F, SE VALE “0” SALTA L’ISTRUZIONE SUCCESSIVA SOSTITUENDOLA CON UN ISTRUZIONE NOP


· ESEMPIO
	
	QUESTA ISTRUZIONE VIENE ESEGUITA CON UN CICLO MACCHINA SE IL BIT ANALIZZATO VALE “1”, ALTRIMENTI SE ESEGUE ANCHE IL NOP VIENE ESEGUITA CON DUE CICLI MACCHINA.


· DATO : BYTE
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	BTFSS


	f,b
	BIT TEST F, SKIP IF SET
	TESTA IL BIT b DEL REGISTRO F, SE VALE “1” SALTA L’ISTRUZIONE SUCCESSIVA SOSTITUENDOLA CON UN ISTRUZIONE NOP


· ESEMPIO
	
	QUESTA ISTRUZIONE VIENE ESEGUITA CON UN CICLO MACCHINA SE IL BIT ANALIZZATO VALE “0”, ALTRIMENTI SE ESEGUE ANCHE IL NOP VIENE ESEGUITA CON DUE CICLI MACCHINA.


· DATO : BYTE E CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	INCFSZ


	f,d
	INCREMENT F, SKIN IF ZERO
	INCREMENTA IL REGISTRO F E PONI  IL RISULTATO IN:

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC
SE A SEGUITO DELL’INCREMENTO IL REGISTO VIENE POSTO A ZERO SALTA L’ISTRUZIONE SEGUENTE SOSTITUENDOLA CON UNA ISTR. NOP, IN QUESTO CASO L’ISTR. VIENE ESEGUITA IN DUE CICLI M. .


· ESEMPIO
	
	


· DATO : BYTE E CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	DECFSZ


	f,d
	DECREMENT F, SKIN IF ZERO
	DECREMENTA IL REGISTRO F E PONI  IL RISULTATO IN:

d=0 copia nell’accumulatore OPPURE d=W

d=1 copia in se stesso OPPURE d=F

W e F sono definiti in una EQU del file P16F877.INC

SE A SEGUITO DEL DECREMENTO IL REGISTO VIENE POSTO A ZERO SALTA L’ISTRUZIONE SEGUENTE SOSTITUENDOLA CON UNA ISTR. NOP, IN QUESTO CASO L’ISTR. VIENE ESEGUITA IN DUE CICLI M. .

	
	
	
	


· ESEMPIO
	
	


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	GOTO


	K
	GO TO
	ESEGUE UN SALTO INCONDIZIONATO ALLA LOCAZIONE K DELLA MEMORIA CODICE, AL POSTO DI DELL’INDIRIZZO SI PUO’ METTERE LA LABEL (ETICHETTA)
VIENE ESEGUITA IN DUE CICLI MACCHINA


· ESEMPIO
	     Org     00H
Loop

     Goto     loop
	Questo programma esegue un loop infinito.


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	CALL


	K
	CALL SOBROUTINE
	CHIAMA LA SOUBRUTINE ALL’INDIRIZZO K  O ALLA LABEL.


· ESEMPIO
	
	QUANDO VIENE ESEGUITA, IL MICRO MEMORIZZA LO STATO DEL PC + 1 (PROGRAM COUNTER) NELLO STACK E SOSTITUISCE IN PC L’INDIRIZZO PER ESEGUIRE LA SOBRUTINE, AL TERMINE RILEGGENDO LO STACK IL PROG. RIPARTE DALL’ISTRUZIONE SUCCESSIVA ALLA CHIAMATA.


· TIPO : CONTROLLO FLUSSO, SALTI E SOBROUTINE  (PARTE 2)
· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	RETURN


	
	RETURN
	TERMINA L’ESECUZIONE DI UNA SUBROUTINE, E FA RIPARTIRE IL PROGRAMMA DALLA ISTRUZIONE SUCCESSIVA A QUELLA DI CHIAMATA DELLA SUBROUTINE.

VIENE ESEGUITA IN DUE CICLI MACCHINA.


· ESEMPIO
	
	


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	RETLW


	K
	RETURN WITH LITERAL IN W
	TERMINA L’ESECUZIONE DI UNA SUBROUTINE, E FA RIPARTIRE IL PROGRAMMA DALLA ISTRUZIONE SUCCESSIVA A QUELLA DI CHIAMATA DELLA SUBROUTINE E PONE K IN W (ACCUMULATORE).

VIENE ESEGUITA IN DUE CICLI MACCHINA.


· ESEMPIO
	
	


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	RETFIE


	
	RETURN FROM INTERRUPT
	TERMINA L’ESECUZIONE DI UNA ROUTINE DI INTERRUPT  E FA RIPARTIRE IL PROGRAMMA DALLA ISTRUZIONE SUCCESSIVA A QUELLA DI CHIAMATA DELLA SUBROUTINE.

VIENE ESEGUITA IN DUE CICLI MACCHINA.


· ESEMPIO
	
	


· TIPO : ISTRUZIONI DI CONTROLLO SISTEMA
· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	CLRWDT


	
	CLEAR WDT
	PONE A “0” I BIT DEL REGISTRO WDT. SE E’ ABILITATO VA ESEGUITO UN CLRWDT PER NON FAR RESETTARE IL MICRO DAL WATCH DOG


· ESEMPIO
	
	


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	SLEEP


	
	SLEEP
	METTE IN STANDBY IL MICRO. PER RIFAR PARTIRE IL MICRO BISOGNA RESETTARLO O UN INTERRUPT SE ABILITATI


· ESEMPIO
	
	


· DATO : CONTROLLO
· FLAG DI STATO MODIFICATI : NESSUNO
· DESCRIZIONE BREVE
	ISTRUZIONE
	PARAMETRI
	TRADUZIONE LETTERALE
	DESCRIZIONE

	NOP


	
	NO OPERATION
	NON FA NULLA, VIENE ESEGUITA IN UN CICLO MACCHINA


· ESEMPIO
	
	


DIAGRAMMI DI FLUSSO

La struttura sequenziale è un insieme (o blocco) di istruzioni da eseguire una dopo l'altra. La struttura condizionale prevede la possibilità di porsi una domanda (contenuta nel rombo) a cui si può rispondere con un si o con un no, e di eseguire istruzioni (o blocchi di istruzioni) differenti a seconda dei due casi. La terza struttura permette di ripetere una o più istruzioni finchè la risposta alla domanda vale si (oppure no a seconda delle necessità). Con questa simbologia grafica è possibile rappresentare il funzionamento di un qualsiasi programma sotto forma di flowchart (diagramma di flusso). Ciascun rettangolo può essere un'istruzione elementare, oppure contenere più istruzioni o anche sottostrutture. Uno dei rettangoli dell'istruzione condizionale può cioè contenere al suo interno altre strutture condizionali e/o iterative, che a loro volta ne possono contenere delle altre... il livello di dettaglio che si vuole rappresentare dipende dalle necessità caso per caso.


Qui a sinistra sono rappresentati i due simboli con cui si inizia e si finisce un                          flowchart. Al posto di inizio e fine oppure sotto spesso viene scritto il nome del programma o della subroutine, infatti ogni sottoprogramma inizia con un blocco di questo tipo con il nome.


Questo blocco contiene una chiamata a una subroutine
Ma vediamo allora come un programma può porsi una domanda grazie alle  istruzioni : BTFSS e BTFSC. Queste servono a testare (o controllare) il valore (stato logico 1 o 0) di un qualsiasi bit di un qualsiasi registro, e permettono di saltare (skip) l'esecuzione dell'istruzione successiva se il bit testato si trova nello stato previsto. La prima salta l'istruzione seguente se il bit testato è 1 (alto, set), la seconda se è 0 (basso, clear). Il bit da testare è specificato con il parametro b, e come sempre può andare da 0 per indicare il bit meno significativo, a 7 per quello più significativo.

Come si vede in questi due esempi viene testato il flag Z, in fondo i flags sono due bit come tutti gli altri, e diventa evidente la loro grandissima importanza. Infatti è anche grazie al loro valore  (che dipende dal risultato delle istruzioni appena eseguite) che il programma può prendere decisioni anche complesse e comportarsi in modi diversi a seconda delle circostanze.

Un semplice skip dell'istruzione successiva non fornisce però ancora la flessibilità necessaria per realizzare una completa struttura condizionale equivalente all' IF THEN ELSE (se, allora altrimenti) dei linguaggi ad alto livello. Questa si ottiene con l'aggiunta dell'istruzione GOTO (vai a, salta), che permette di saltare ad un punto qualsiasi del programma da un qualsiasi altro punto. Quando una GOTO è usata insieme ad una istruzione di skip si crea un salto condizionato. E' naturalmente possibile mettere una GOTO in qualsiasi punto del programma, in tal caso questo salto prende il nome di salto incondizionato. E' buona norma però ridurre il più possibile l'utilizzo dei salti perché rendono complicata la lettura e la manutenzione/correzione del programma. Il loro utilizzo andrebbe limitato (se possibile) esclusivamente per la creazione delle strutture fondamentali condizionale e iterativa. Per indicare il punto di arrivo di un salto (GOTO) si usano normalmente delle etichette (label), cioè dei nomi inizianti nella prima colonna del testo. E' evidente che in un programma non devono mai esserci due label uguali, in quanto ciascuna identifica un indirizzo della memoria programma ben preciso.

L'esempio seguente mostra come realizzare in assembler la struttura IF THEN ELSE completa. Nell'esempio viene testato il bit 0 del registro DIREZ, se vale 0 viene eseguita l'istruzione RLF LUCE,F (saltando la GOTO ELSE) seguita dal salto a ENDIF, altrimenti viene eseguita l'istruzione RRF LUCE,F. Le istruzioni da eseguire in un caso o nell'altro possono essere anche più di una, a differenza della skip semplice che al massimo permette di saltare l'esecuzione di una singola istruzione.

L' esempio seguente presenta  la struttura iterativa, ed è molto importante comprenderlo appieno perché su questo genere di struttura/funzionamento si basa gran parte dell'elaborazione nei casi reali.

E' una generica sezione di programma in cui ad un certo punto si carica il valore 4 in un registro della RAM (chiamato come al solito PIPPO).  Seguono poi delle istruzioni indicate con dei puntini. Ed infine troviamo altre 4 istruzioni. Le prime due servono a sottrarre 1 dal registro PIPPO. La terza controlla il valore del flag Z che viene settato quando il risultato dell'istruzione precedente vale 0. Se il flag Z è settato  viene saltata l'istruzione seguente e il programma prosegue. Altrimenti, se PIPPO dopo la sottrazione non vale 0 e il flag Z non è settato, lo skip non viene effettuato, perciò viene eseguita la GOTO che rimanda indietro al punto indicato dall'etichetta NEXT. In pratica si crea quello che viene chiamato un loop (ciclo, anello) che ripete il blocco di istruzioni interne esattamente 4 volte. Il registro PIPPO funziona da contatore di ciclo. Siccome il suo valore parte da 4 e decrementa ad ogni ciclo, al quarto passaggio raggiunge lo 0 e il loop termina, il programma prosegue cioè con le istruzioni successive....

       MOVLW   4         ;W=4

       MOVWF   PIPPO     ;PIPPO=W

NEXT   ....

       ....

       MOVLW   1         ;W=1

       SUBWF   PIPPO,F   ;PIPPO=PIPPO-W

       BTFSS   STATUS,Z  ;SE PIPPO=0 ALLORA SKIP (FINE CICLO)

       GOTO    NEXT      ;ALTRIMENTI TORNA A NEXT

Abbiamo ormai quasi tutti gli elementi per poter finalmente animare un pò i nostri LED, prima però vediamo qualche altra istruzione. La terza e la quarta istruzione della tabella sono istruzioni potenti, in quanto permettono in un colpo solo di incrementare o decrementare di 1 un registro, ed effettuare automaticamente uno skip dell'istruzione seguente se il risultato dell'operazione vale 0. L'esempio precedente può perciò essere riscritto più succintamente nel seguente modo:

       MOVLW   4         ;W=4

       MOVWF   PIPPO     ;PIPPO=W

NEXT   ....

       ....

       DECFSZ  PIPPO,F   ;DECR.PIPPO, SE=0 ALLORA SKIP (FINE CICLO)

       GOTO    NEXT      ;ALTRIMENTI TORNA A NEXT
Con i salti (condizionati e non) è già possibile creare qualsiasi flusso elaborativo, tuttavia se c'è bisogno di creare delle suddivisioni logiche tra sezioni di programma che eseguono  compiti specifici, oppure ci sono gruppi di istruzioni ripetuti frequentemente in diversi punti del programma, è preferibile ricorrere alle subroutines (o sottoprogrammi). Ogni subroutine può essere così considerata come un modulo a parte, da chiamare quando serve il suo servizio. Le subroutines sono semplicemente porzioni di programma che vengono chiamate con l'istruzione CALL (nello stesso modo di GOTO) ma terminano con l'istruzione RETURN che fa ritornare il programma all'istruzione successiva alla CALL di partenza. Questo comportamento è possibile perché ad ogni CALL il PIC salva il contatore di programma (il program counter, che tiene conto dell'indirizzo dell'istruzione in esecuzione) in un'area speciale chiamata stack, e lo riprende quando incontra una RETURN. Il PIC 16F628 ha un'area di stack a 8 livelli, è quindi possibile chiamare fino a 8 subroutines una dentro l'altra (nidificate o nested). L'uso delle subroutines è estremamente consigliato, in quanto sono l'unica cosa in grado di tenere in ordine un programma complesso spezzandolo in più moduli, cioè porzioni di codice di dimensioni ridotte che eseguono ciascuna un compito ben specifico, più facile da comprendere e da mettere a punto. Un'altra buona ragione per l' utilizzo dei sottoprogrammi è che all'evenienza possono essere copiati in altri programmi per essere riutilizzati. In gergo una raccolta utile di subroutines riutilizzabili si chiama libreria.

Ci sono altre due forme di istruzione di ritorno da subroutine, la RETLW e la RETFIE. La prima permette il ritorno da una subroutine con un valore specifico caricato nell'accumulatore, come vedremo serve per creare delle tabelle dati nell'area programma, cosa altrimenti impossibile in quanto questa memoria non è accessibile dalle comuni istruzioni che lavorano solo sui registri dalla memoria dati. La RETFIE invece si usa nel caso in cui la subroutine da terminare sia un gestore di interrupt.

Esperimento:
Ed ora, per mettere in pratica tutto quanto, subroutines, salti condizionati e rotazioni di bit, vogliamo ottenere sui LED  un effetto supercar, cioè deve accendersi un LED alla volta scorrendo da destra verso sinistra per poi tornare indietro e così via.

Dapprima definiamo le variabili di lavoro (registri) che ci occorrono. Un registro di nome LUCE conterrà il valore a 8 bit da scrivere sulla porta di uscita, uno chiamato DIREZ conterrà 0 o 1 a seconda che la direzione di scorrimento voluta sia verso sinistra o verso destra, ed infine due registri di nome L_CONT e H_CONT serviranno come contatore di ciclo a 16 bit per una subroutine di ritardo. Quest'ultima è necessaria perché la velocità di esecuzione sarebbe altrimenti così rapida da non poter essere notata, in pratica tutti i LED apparirebbero ugualmente illuminati a causa della persistenza dell'immagine sulla nostra retina. Visto che ogni istruzione impiega un certo tempo per essere eseguita è evidente che eseguendo decine di migliaia di volte poche istruzioni all'interno di un loop si possono ottenere ritardi di tempo considerevoli. E' necessario usare 16 bit per il conteggio in quanto usandone solo 8 il massimo ritardo ottenibile sarebbe troppo breve (meno di 2 millisecondi). Per visualizzare un effetto gradevole ci occorre invece un ritardo di diverse decine di mS. Una soluzione poteva essere quella di richiamare la sbroutine di ritardo a 8 bit della durata di 2mS molte volte di fila, ma questo avrebbe in ogni caso comportato l'uso di un altro ciclo e di un altro registro di conteggio. In questo esempio invece si usano due bytes in modo tale da considerarli come un unico registro a 16 bit, con cui realizzare cicli da 1 a 65536 ripetizioni, ed è il primo esempio di creazione via software di un tipo di dato che può assumere valori maggiori di quelli trattabili direttamente dal micro. Un valore a 16 bit è composto infatti da due bytes detti alto (H, più significativo) e basso (L, meno significativo). Il valore contenuto in quello alto ha peso 256, una unità nel byte alto vale cioè 256 volte la stessa unità nel byte basso, questo vuol dire che il valore complessivo a 16 bit di due bytes H ed L è dato da H*256 + L. Nel nostro caso impostiamo un ciclo di 5320 ripetizioni, la parte alta varrà 20 e quella bassa 200 (infatti 20*256+200=5320). Per decrementare un valore a 16 bit si decrementa prima la parte bassa, se si ha un rollover negativo (da 0 si torna a 255) allora si decrementa anche la parte alta. Quando entrambe le parti contengono zero (basta fare un OR tra di loro e verificare se si setta il flag Z) il ciclo termina. Durante l'esecuzione del programma il valore binario iniziale 00000001 di LUCE viene spostato verso sinistra di una posizione alla volta tramite l'istruzione RLF. L'azzeramento del flag C prima della rotazione fa si che da destra non entri mai un altro bit a 1. Quando il bit a 1 contenuto in LUCE raggiunge la posizione 7 (cioè l'ottava a sinistra, la piùsignificativa) il valore del registro direzione viene messo a 1, in modo tale da abilitare lo spostamento verso destra. Allo stesso modo quando il bit  a 1 di LUCE ritorna nella posizione 0 il valore della direzione cambia di nuovo riabilitando lo spostamento verso sinistra come all'inizio.  Il programma è senza fine perché il GOTO finale rimanda all'infinito alla posizione MAINLOOP, in totale occupa 33 locazioni di memoria programma delle 2048 disponibili, e utilizza 4 bytes di memoria RAM dei 224 disponibili.

;-----------------------------------------------------

; Programma effetto supercar

;-----------------------------------------------------

          PROCESSOR 16F628

          RADIX     DEC

          INCLUDE   "P16F628.INC"

          __CONFIG  11110100010000B

;-----------------------------------------------------

LUCE      EQU       32

H_CONT    EQU       33

L_CONT    EQU       34

DIREZ     EQU       35

;-----------------------------------------------------

          ORG       0

          BSF       STATUS,RP0  ;Attiva banco 1

          CLRF      TRISB       ;Rende PORTB un'uscita

          BCF       STATUS,RP0  ;Ritorna al banco 0

          MOVLW     00000001B

          MOVWF     LUCE        ;LUCE=00000001

          CLRF      DIREZ       ;DIREZ=0

MAINLOOP  MOVF      LUCE,W

          MOVWF     PORTB       ;Scrive LUCE sulla PORTB

          BCF       STATUS,C    ;Azzera flag C

          BTFSC     DIREZ,0     ;Se DIREZ=0 (sinistra) skip

          GOTO      LAB1        ;altrimenti GOTO LAB1

          RLF       LUCE,F      ;Ruota LUCE verso sinistra

          GOTO      LAB2        ;GOTO fine struttura IF

LAB1      RRF       LUCE,F      ;Ruota LUCE verso destra

LAB2      BTFSC     LUCE,7      ;Se bit 7 di LUCE è 0 skip

          INCF      DIREZ,F     ;altrimenti direzione=destra

          BTFSC     LUCE,0      ;Se bit 0 di luce è 0 skip

          CLRF      DIREZ       ;altrimenti direzione=sinistra

          CALL      DELAY       ;Richiama subroutine di ritardo

          GOTO      MAINLOOP    ;Nuovo ciclo del programma

;-----------------------------------------------------

DELAY     MOVLW     20          ;Carica 5320 nei 16 bit

          MOVWF     H_CONT      ;formati dai due bytes

          MOVLW     200         ;H_CONT e L_CONT

          MOVWF     L_CONT

DELAY2    DECF      L_CONT,F    ;Decrementa parte bassa del contatore

          COMF      L_CONT,W    ;Inverte i bit

          BTFSC     STATUS,Z    ;Se tutti zero c'è stato rollover

          DECF      H_CONT,F    ;allora decrementa parte alta

          MOVF      L_CONT,W    ;Carica in W la parte bassa

          IORWF     H_CONT,W    ;Mettila in OR con la parte alta

          BTFSS     STATUS,Z    ;Se tutto zero skip (fine ciclo)

          GOTO      DELAY2      ;Altrimenti ritorna a DELAY2

          RETURN

;-----------------------------------------------------

          END

Qui sotto sono riportati i flowchart completi della sezione principale del programma (chiamata anche main) e della subroutine delay. A fianco del disegno sono riportate le istruzioni che compongono le varie parti. Come si può vedere la chiamata alla subroutine si rappresenta con un rettangolo con aggiunte due righe laterali.

Il programma è completo e funzionante, tuttavia è possibile una piccola ottimizzazione della struttura IF THEN ELSE.  Infatti nel caso cui vi sia una sola istruzione da eseguire sotto il THEN e una sola sotto l'ELSE la struttura può essere scritta in modo più compatto risparmiando una istruzione e la necessità di dover definire due etichette a cui saltare, da così:

          BTFSC     DIREZ,0

          GOTO      LAB1

          RLF       LUCE,F

          GOTO      LAB2

LAB1      RRF       LUCE,F

LAB2      ......

a così: 

          BTFSC    DIREZ,0

          RRF      LUCE,F

          BTFSS    DIREZ,0

          RLF      LUCE,F

Il main del nostro programma supercar diventerebbe cioè: 

          BSF       STATUS,RP0  ;Attiva banco 1

          CLRF      TRISB       ;Rende PORTB un'uscita

          BCF       STATUS,RP0  ;Ritorna al banco 0

          MOVLW     00000001B

          MOVWF     LUCE        ;LUCE=00000001

          CLRF      DIREZ       ;DIREZ=0

MAINLOOP  MOVF      LUCE,W

          MOVWF     PORTB       ;Scrive LUCE sulla PORTB

          BCF       STATUS,C    ;Azzera flag C

          BTFSC     DIREZ,0     ;Se DIREZ=0 (sinistra) skip

          RRF       LUCE,F      ;Ruota LUCE verso destra

          BTFSS     DIREZ,0     ;Se DIREZ=1 (destra) skip

          RLF       LUCE,F      ;Ruota LUCE verso sinistra
          BTFSC     LUCE,7      ;Se bit 7 di LUCE è 0 skip

          INCF      DIREZ,F     ;altrimenti direzione=destra

          BTFSC     LUCE,0      ;Se bit 0 di luce è 0 skip

          CLRF      DIREZ       ;altrimenti direzione=sinistra

          CALL      DELAY       ;Richiama subroutine di ritardo

          GOTO      MAINLOOP    ;Nuovo ciclo del programma

Lo stesso sistema può essere usato per spostare o meglio copiare un singolo bit di un qualsiasi registro in un altro bit di un qualsiasi altro registro. Nell'esempio seguente copiamo il valore del flag C nel bit 4 del registro PIPPO:

         BTFSC    STATUS,C

         BSF      PIPPO,4

         BTFSS    STATUS,C

         BCF      PIPPO,4

La sequenza appena vista va bene anche per trasferire il valore di un singolo bit della RAM su un singolo pin di uscita di una porta. Se si lavora invece solo con registri in memoria, come in questo questo caso specifico, è possibile in realtà una ulteriore riduzione di istruzioni. Basta impostare inizialmente il bit di PIPPO ad un valore specifico, per esempio 0, e poi cambiandolo subito dopo SOLO SE il flag C è settato:

        BCF      PIPPO,4

        BTFSC    STATUS,C

        BSF      PIPPO,4

Per concludere va ricordato che tutte le istruzioni viste in precedenza duravano sempre un ciclo macchina. Le istruzioni viste in questo capitolo invece possono durare anche 2 cicli macchina. Per la precisione le istruzioni di salto, chiamata e ritorno (GOTO, CALL, RETURN, RETLW, RETFIE) richiedono sempre 2 cicli macchina, mentre gli skip ne richiedono 2 solo se la condizione è verificata.

